
 Simulating a mixture of normal variables

Many computer programs offer a possibility of creating data that can, at least for most practical purposes, be considered random from a certain probability distribution. However, not seldom there is a need to create data from non-standard distributions. There are many ways of doing this, from rather simply ones to complicated or very complicated ways, as there are sometimes some very special needs on the data. (See also %Mix.)
The example below can be a bit messy especially as you do not have all the details and the expla​nations are insufficient (it would be necessary to include and perhaps rewrite the whole article from were it comes). See it just as an example of a mixture found in describing the process of wafer manufacturing. (Run also the macro %Mix in order to get more info about mixtures.)

A man wanted to simulate data from a fairly complicated distri​bution which, after taking a look at the function, turned out to consist of a mixture of normal distri​butions. Before doing the simulation we want to investigate the distribution (M is the number of machines, a is a probability i.e. a value in the range [0, 1]. A shift in mean is expressed in S sigma):

Is this a proper distribution? We always have two requirements: 

1. the f(x) must be positive, i.e. above the X-axis

2. the total area under the curve must be 1

The first requirement is fulfilled as it is a sum of the function of normal distributions with positive coefficients, i.e. always positive and thus always above the X-axis.

If we integrate both sides of the equal sign, we get the total area under the curve f(x). And as usual, integration of a probability from minus to plus infinity gives the area 1. We get the following:

Thus requirement 2 is also fulfilled.

An extra comment.  If the probability a = 0, then the distribution will consist of the third term only. If a = 1 the distribution will consist of the first two terms only. If M = 1 he distribution will consists of the first and last terms (depending on the value of a). NB that we do not discuss the meaning of the probability a.'

The simulation.  The simulation is straight forward. In the commands below we use constants for the 'unknowns' M, a, S, ,  and simulates the three distributions into three different columns. After this, for each row we chose one of the columns according to the three probabilities 

. These are stored in column c21. Copy all the rows below and paste them into the session window of Minitab. This will draw the distribution. Change some of the unknowns and see the distribution change.

The origin of the distribution comes from a somewhat technical paper discussing a process chart.

# Simulating a mixture of normal distributions.

erase c1-c4000

erase k1-k1000

name c1 'X-axis' c2 'pdf 1' c3 'pdf 2' c4 'pdf 3'

name c6 'Mixed pdf' c8 'Data 1' c9 'Data 2' c10 'Data 3'

name c11 'Term' c13 'Mixed data'

name k31 'tot mean' k32 'tot std'

# ------------------------------------ Calculation of f(x)

let k1 = 3                         # M.

let k2 = 0.3                       # a.

let k3 = 3.2                       # S.

let k4 = 50                        # mu.

let k5 = 5                         # sigma.

let k6 = k4 + k3*k5                # mu of 1st distribution.

let c30(1) = k6                    # To choose min and

let c30(2) = k4                    # max expected values.

let k10 = min(c30)

let k11 = max(c30)

let k12 = k10 - 3.5*k5             # min.point on X-axis.

let k13 = k11 + 3.5*k5             # max.point on X-axis.

let k14 = (k13 - k12)/100          # Step size of X-axis.

set c1                             # Creates the X-axis.

k12:k13/k14

end

pdf c1 c2;                         # pdf 1st distribution.

normal k6 k5.

pdf c1 c3;                         # pdf 2nd distribution.

normal k4 k5.

pdf c1 c4;                         # pdf 3rd distribution.

normal k4 k5.

                                   # mixed pdf

let c6 = k2/k1*c2 + (1 - 1/k1)*k2*c3 + (1 - k2)*c4

# -----------------------------------------------------

# ------------------------------------ Simulation of f(x)

let k21 = 10000                    # Number of data values.

random k21 c8;                     # Simulation from the 1st distribution.

normal k6 k5.

random k21 c9;                     # Simulation from the 2nd distribution.

normal k4 k5.

random k21 c10;                    # Simulation from the 3rd distribution.

normal k4 k5.

let c21(1) = k2/k1                 # Weights for the three distributions.

let c21(2) = (1 - 1/k1)*k2

let c21(3) = (1 - k2)

set c22

1 2 3

end

rand k21 c11;                      # 1, 2 or 3 to choose distribution.

discrete c22 c21.

let c13 = (c11=1)*c8 + (c11=2)*c9 + (c11=3)*c10

layout;

graph;

etype 1;

esize 1;

type 1;

color 46;

wtitle 'Simulation of a mixed distribution';

title 'Simulation of a mixed distribution';

tsize 1.1;

title 'Histogram: the simulated data.  Curve: theoretically derived pdf.';

tsize 0.9;

tcolor 4;

text 0.02 0.015 '(c) Ing-stat, www.ing-stat.nu';

tsize 0.5;

text 0.06 0.95 "For this mixture the following is valid:";

tsize 0.6;

tcolor 4;

text 0.10 0.91 "mt = b1*m1 + b2*m2 + b3*m3";

tsize 0.6;

tcolor 4;

text 0.10 0.88 "vt = b1*(v1 + (m1-mt)**2) + b2*(v2 + (m2-mt)**2)+ b3*(v3 + (m3-mt)**2)";

tsize 0.6;

tcolor 4;

text 0.06 0.84 "mt = total mean value,  vt = total variance,  bi = prop;  Sum(bi) = 1";

tsize 0.6;

tcolor 2;

text 0.06 0.81 "Run e.g. the %Ndata or %Disttest macros to investigate the data.";

tsize 0.6;

tcolor 2;

text 0.06 0.78 "Run also the macro %Mix that shows more about mixing of variables.";

tsize 0.6.

hist c13;

Scale 1;

tsize 0.6;

length 0.005;

HDisplay 0 0 0 0;

Scale 2;

length 0.005;

tsize 0.6;

HDisplay 0 0 0 0;

LDisplay 0 0 0 0;

nodt;

line c1 c6;

size 2;

marker k6 0;

type 29;

color 14;

size 1.5;

marker k6 0;

type 28;

size 1.5;

marker k4 0;

type 29;

color 14;

size 1.5;

marker k4 0;

type 28;

size 1.5;

bar;

type 1;

color 15;

data 0.15 0.85  0.09 0.77;

etype 0;

density;

axlabel 1 '';

axlabel 2 ''.

endlayout

let k31 = c21(1)*k6 + c21(2)*k4 + c21(3)*k4    # Tot mu

let k32 = sqrt(c21(1)*(k5**2 + (k6-k31)**2) +         &

               c21(2)*(k5**2 + (k4-k31)**2) +         &

               c21(3)*(k5**2 + (k4-k31)**2))   # Tot sigma

print k31 k32                      # Prints mu, sigma.

descr c13                          # Stats from data.

